

Computer Assistance with Math Fact Woes

A. Craig Loewen

Rosalind School. County of Camrose

The use of computers is increasing in Alberta schools and classrooms. Teachers are left with the task of determining exactly how computers may best suit their own and their students' needs. This is an individual decision. One inexpensive, yet effective, computer application in a combined Grade 4 and 5 mathematics class is explored.

Sandra knows her multiplication and division facts perfectly. Matthew performs adequately as long as there are no factors or divisors greater than seven. Pat struggles with everything. After having worked through all the appropriate and specified exercises and activities, it became obvious that some students required remedial exercises while others would best benefit from occasional drills and exercises to ensure the retention of acquired skills. There are probably many ways of overcoming this difficulty, but the computer could be helpful. A program was written that would print computer generated mathematics fact drill sheets.

The drill sheets allow for individual differences in the students. To meet Sandra's needs, the computer could quickly print out 100 facts using all the factors 0 through 10. Just as quickly, Matthew could have a collection of questions concentrating on 8, 9, and 10 as factors. Pat could receive a series of sheets which would progressively drill her in all her weak areas. The program creates drills molded to student needs in a fraction of the time required to assemble comparable worksheets by hand or typewriter. If necessary, a single sheet can be created and duplicated for use as a class set during a timed drill or other full class exercise.

Another advantage of the program, is that unlike flashcards, it provides something tangible the students can return to the teacher. If sheets are taken home at night and returned the next day, it is easy to chart progress and areas of weakness. Flashcards seem to provide little feedback to the teacher. Identified trouble areas can be drilled again and again as each sheet the computer prints is different from the last.

The program that follows is made to be as simple as possible. It will run on any Apple II system with a printer connected in slot #1. The program asks the teacher to input minimum and maximum values for the factors. The computer does the rest. The program will print out 100 questions with factors in the range specified. When running the program, you will see something like this:

```
THIS PROGRAM GENERATES MATH WORKSHEETS
FIRST FACTOR--LOWEST VALUE: 0
FIRST FACTOR--HIGHEST VALUE: 12
SECOND FACTOR--LOWEST VALUE: 0
SECOND FACTOR--HIGHEST VALUE: 12
GENERATING WORKSHEET
```

The worksheet on the printer will have this format:

MATH FACTS DRILL
FILL IN EACH BLANK WITH THE CORRECT VALUE.

12 X 1 = _____ 0 X 10 = _____ 7 X 8 = _____ 12 X 7 = _____
10 X 10 = _____ 1 X 10 = _____ 11 X 10 = _____ 7 X 3 = _____

To make a nicer looking page for the class sets, you could beautify the sheet with clip art before duplicating.

A copy of the program follows. Before you start typing, you should know that there are other sources for similar programs. MECC has developed a package called "Mastering Math Worksheet Generator," but it is oriented specifically to the Mastering Math text series. It is available through ACCESS. Some commercial packages are available, but these can be quite expensive (\$60 to \$200), depending on the number of different types of worksheets the package is able to generate.

The Program

```
10 REM MATH DRILL SHEET GENERATOR
20 REM JANUARY 1986
30 HOME: DIM Q$(100), A$(100)
40 PRINT "THIS PROGRAM GENERATES MATH WORKSHEETS"
50 INPUT "FIRST FACTOR--LOWEST VALUE: ";FL
60 INPUT "FIRST FACTOR--HIGHEST VALUE: ";FH
70 IF FL > FH OR FH > 99 THEN PRINT "INPUT ERROR": GOTO 50
80 INPUT "SECOND FACTOR--LOWEST VALUE: ";SL
90 INPUT "SECOND FACTOR--HIGHEST VALUE: ";SH
100 IF SL > SH OR SH > 99 THEN PRINT "INPUT ERROR": GOTO 80
110 PRINT "GENERATING WORKSHEET"
120 FOR X = 1 TO 100
130 A = INT ( RND (1) * (FH - FL + 1) + FL)
140 B = INT ( RND (1) * (SH - SL + 1) + SL)
150 Q$(X) = RIGHT$ (" " + STR$(A),3) + " X" + RIGHT$ (" " + STR$(B),3) +
"= _____"
160 A$(X) = LEFT$ ( LEFT$ (Q$(X),12) + STR$(A * B) + "-----", 16)
170 NEXT X
180 PR# 1
190 PRINT CHR$(13); CHR$(13)"MATH FACTS DRILL"
200 PRINT "FILL IN EACH BLANK WITH THE CORRECT VALUE."; CHR$(13); CHR$(13)
210 FOR X = 1 TO 100
220 PRINT Q$(X) " ";
230 IF X / 4 = INT (X / 4) THEN PRINT CHR$(13)
240 NEXT X
250 FOR X = 1 TO 7: PRINT CHR$(10): NEXT
260 PRINT "MATH FACTS DRILL -- ANSWER KEY" CHR$(13); CHR$(13)
270 FOR X = 1 TO 100
```

```

280 PRINT A$(X) " ";
290 IF X / 4 = INT (X / 4) THEN PRINT CHR$( 13)
300 NEXT X
310 END

```

In order to make worksheets for the other three operations you will have to change these lines:

ADDITION:

```

50 INPUT "FIRST ADDEND--LOWEST VALUE: ";FL
60 INPUT "FIRST ADDEND--HIGHEST VALUE: ";FH
80 INPUT "SECOND ADDEND--LOWEST VALUE: ";SL
90 INPUT "SECOND ADDEND--HIGHEST VALUE: ";SH
150 Q$(X) = RIGHT$( " " + STR$( A),3) + " + " + RIGHT$( " " + STR$( B),3) +
" = ----"
160 A$(X) = LEFT$( ( LEFT$( Q$(X),12) + STR$( A + B) + "----",16)

```

SUBTRACTION:

```

50 INPUT "DIFFERENCE--LOWEST VALUE: ";FL
60 INPUT "DIFFERENCE--HIGHEST VALUE: ";FH
80 INPUT "SUBTRAHEND--LOWEST VALUE: ";SL
90 INPUT "SUBTRAHEND--HIGHEST VALUE: ";SH
150 Q$(X) = RIGHT$( " " + STR$( A + B),3) + " -" + RIGHT$( " " + STR$( B),3) +
" = ----"
160 A$(X) = LEFT$( ( LEFT$( Q$(X),12) + STR$( A) + "----",16)

```

DIVISION:

```

50 INPUT "QUOTIENT--LOWEST VALUE: ";FL
60 INPUT "QUOTIENT--HIGHEST VALUE: ";FH
80 INPUT "DIVISOR--LOWEST VALUE: ";SL
90 INPUT "DIVISOR--HIGHEST VALUE: ";SH
150 Q$(X) = RIGHT$( " " + STR$( A * B ),3) + " %" + RIGHT$( " " + STR$( B),3)
+ " = ----"
160 A$(X) = LEFT$( ( LEFT$( Q$(X),12) + STR$( A) + "----",16)

```

Now, simply by running any of these four programs, the teacher can create individualized worksheets or class drill sheets without the previously required time and work. In a split elementary mathematics class, time is valuable, but so is the students' learning. With this program, neither has to be sacrificed.

A. Craig Loewen teaches mathematics and computing science courses at Rosalind School in the County of Camrose. Craig was the County of Camrose nominee for the Alberta School Trustees Edwin Parr Award in 1985.